
 
 
Name   GEY 715 Fall 2003 Homework No. 3 
 

Basic Aquifer Simulation Using a Two-Dimensional Groundwater Model 
 
Introduction 

In GEY 674--Hydrogeology you were introduced to the concepts of well hydraulic and 
various analytic and graphic methods for estimating drawdowns and their effect on the water table. 
The purpose of this assignment is to learn how to prepare input data sets, execute the program, and 
analyze modeling results. 
 
Problem No. 1 

Using the FORTRAN compiler on the UNIXs (or PCs) in the Hydrology Computing Lab, 
copy the code (pltest.for) located in /home1/gey715/ to your working directory and run the 
program. Figure 2 in Yu's paper (1997) illustrates the parameters used for the simulation in this 
assignment. The program is used here for analyzing cause and effect relationship involving 
drawdowns or hydraulic heads in a non-steady-state, nonhomogeneous and/or homogeneous, 
isotropic aquifer system under nonleaky unconfined aquifers.  
 

The two-dimensional flow problem in Figure 2 will be modeled with uniform grid spacing 
(31x31), homogeneous and isotropic aquifer system properties, initial heads set at the zero 
reference level, and equal time increments (or you can use unequal time increments). A pumping 
well needs to be set in grid block (16,16) in an unconfined aquifer with a grid of 31 columns and 31 
rows. First, you need to benchmark the code by running the model and re-producing Figure 3. 
Figure 3 contains the simulated results and estimates through Theis curve method at t=21 days. 
Then conduct additional simulations to answer following questions. 
 
1. With current parameters, what are the drawdowns at times of 1, 5, and 10 days.  
2. How would the change in K value affect the drawdowns (with simulation runs and results)? 
3. How would the thickness affect the drawdowns (with simulation runs and results)? 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



 
Preparing the input file 
 
      READ(5,10)NSTEPS,DELTA,ERROR, 
     1NC,NR,TT,S1,HH,QQ 
10    FORMAT(I6,2F6.0/2I6,4F6.0) 
 
where  NSTEPS: number of time steps   (42) 
 DELTA: time increment    (0.5) 
 ERROR: convergence error    (1.0) 
 NC: number of columns    (31) 
 NR: number of rows      (31) 
 TT: transmissivity (gpg/ft)     (74800) 
 S1: storativity factor S1=6.23 (delta x) (delta y) (75000) 
 HH: initial water level     (0.0) 
 QQ: recharge      (0.0) 
 
30    READ(5,40,END=50)I,J,T(I,J,1), 
     1T(I,J,2),SF1(I,J),H(I,J),Q(I,J) 
40    FORMAT(2I3,2F6.0,2F4.0,1F6.0) 
 
where  
 I: number of columns     (16) 
 J: number of rows      (16) 
 TT: transmissivity (gpg/ft)     (74800) 
 S1: storativity factor S1=6.23 (delta x) (delta y) (75000) 
 HH: initial water level     (0.0) 
 QQ: recharge      (1E6) 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



C          PRICKETT AND LONNQUIST (1971)   
C          BASIC AQUIFER SIMULATION PROGRAM 
C          (MODIFICATIONS - G. MCCLYMONT) 
      IMPLICIT REAL*8 (A-H,O-Z) 
      DIMENSION H(40,40),HO(40,40), 
     1SF1(40,40),Q(40,40),T(40,40,2), 
     2B(40),G(40),DL(40,40) 
C 
C          DEFINE INPUT AND OUTPUT DEVICE NUMBERS 
C 
      CHARACTER*20 INFILE 
      CHARACTER*20 OUTFILE 
      WRITE(*,900) 
  900 FORMAT (' INPUT FILE NAME- ') 
      READ(*,910) INFILE 
  910 FORMAT (A) 
      WRITE(*,920) 
  920 FORMAT(' OUTPUT FILE NAME-') 
      READ(*,910) OUTFILE 
      OPEN(5,FILE=INFILE) 
      OPEN(6,FILE=OUTFILE,STATUS='NEW') 
C 
C          READ PARAMETER CARD AND 
C          DEFAULT VALUE CARD 
C 
      READ(5,10)NSTEPS,DELTA,ERROR, 
     1NC,NR,TT,S1,HH,QQ 
10    FORMAT(I6,2F6.0/2I6,4F6.0) 
      WRITE(6,743)NSTEPS,DELTA,ERROR,NC,NR,TT,S1,HH,QQ 
743   FORMAT(I10,' : NSTEPS'/F10.2,' : DELTA'/F10.2,' : ERROR'/ 
     1       I10,' : NC'/I10,' : NR'/F10.2,' : TT'/F10.2,' : SF1'/ 
     2       F10.2,' : HH'/F10.2,' : QQ'//) 
C 
C          FILL ARRAYS WITH DEFAULT VALUES 
C 
      DO 20 I=1,NC 
      DO 20 J=1,NR 
      T(I,J,1)=TT 
      T(I,J,2)=TT 
      SF1(I,J)=S1 
      H(I,J)=HH 
      HO(I,J)=HH 
20    Q(I,J)=QQ 
C 
C          READ NODE CARDS 
C 



      WRITE(6,753) 
753   FORMAT('    I    J         T1         T2        SF1        H', 
     1      '         Q') 
30    READ(5,40,END=50)I,J,T(I,J,1), 
     1T(I,J,2),SF1(I,J),H(I,J),Q(I,J) 
40    FORMAT(2I3,2F6.0,2F4.0,1F6.0) 
      WRITE(6,754)I,J,T(I,J,1),T(I,J,2),SF1(I,J),H(I,J),Q(I,J) 
754   FORMAT(I5,I5,1X,E10.3,1X,E10.3,1X,E10.3,2F10.2) 
      GO TO 30 
C 
C          START OF SIMULATION 
C 
50    TIME=0 
      DO 320 ISTEP=1,NSTEPS 
      TIME=TIME+DELTA 
C 
C          PREDICT HEADS FOR NEXT 
C          TIME INCREMENT 
C 
      DO 70 I=1,NC 
      DO 70 J=1,NR 
      D=H(I,J)-HO(I,J) 
      HO(I,J)=H(I,J) 
      F=1.0 
      IF(DL(I,J).EQ.0.0)GO TO 60 
      IF(ISTEP.GT.2)F=D/DL(I,J) 
      IF(F.GT.5)F=5.0 
      IF(F.LT.0.0)F=0.0 
60    DL(I,J)=D 
70    H(I,J)=H(I,J)+D*F 
C 
C          REFINE ESTIMATES OF HEADS BY IADI METHOD 
C 
      ITER=0 
80    E=0.0 
      ITER=ITER+1 
C 
C          COLUMN CALCULATIONS 
C 
      DO 190 II=1,NC 
      I=II 
      IF(MOD(ISTEP+ITER,2).EQ.1) I=NC-I+1 
      DO 170 J=1,NR 
C 
C          CALCULATE B AND G ARRAYS 
C 



      BB=SF1(I,J)/DELTA 
      DD=HO(I,J)*SF1(I,J)/DELTA-Q(I,J) 
      AA=0.0 
      CC=0.0 
      IF(J-1)90,100,90 
90    AA=-T(I,J-1,1) 
      BB=BB+T(I,J-1,1) 
100   IF(J-NR)110,120,110 
110   CC=-T(I,J,1) 
      BB=BB+T(I,J,1) 
120   IF(I-1)130,140,130 
130   BB=BB+T(I-1,J,2) 
      DD=DD+H(I-1,J)*T(I-1,J,2) 
140   IF(I-NC)150,160,150 
150   BB=BB+T(I,J,2) 
      DD=DD+H(I+1,J)*T(I,J,2) 
160   W=BB-AA*B(J-1) 
      B(J)=CC/W 
170   G(J)=(DD-AA*G(J-1))/W 
C 
C          RE-ESTIMATE HEADS 
C 
      E=E+ABS(H(I,NR)-G(NR)) 
      H(I,NR)=G(NR) 
      N=NR-1 
180   HA=G(N)-B(N)*H(I,N+1) 
      E=E+ABS(HA-H(I,N)) 
      H(I,N)=HA 
      N=N-1 
      IF(N)190,190,180 
190   CONTINUE 
C 
C          ROW CALCULATIONS 
C 
      DO 300 JJ=1,NR 
      J=JJ 
      IF(MOD(ISTEP+ITER,2).EQ.1) J=NR-J+1 
      DO 280 I=1,NC 
      BB=SF1(I,J)/DELTA 
      DD=HO(I,J)*SF1(I,J)/DELTA-Q(I,J) 
      AA=0.0 
      CC=0.0 
      IF(J-1)200,210,200 
200   BB=BB+T(I,J-1,1) 
      DD=DD+H(I,J-1)*T(I,J-1,1) 
210   IF(J-NR)220,230,220 



220   DD=DD+H(I,J+1)*T(I,J,1) 
      BB=BB+T(I,J,1) 
230   IF(I-1)240,250,240 
240   BB=BB+T(I-1,J,2) 
      AA=-T(I-1,J,2) 
250   IF(I-NC)260,270,260 
260   BB=BB+T(I,J,2) 
      CC=-T(I,J,2) 
270   W=BB-AA*B(I-1) 
      B(I)=CC/W 
280   G(I)=(DD-AA*G(I-1))/W 
C 
C          RE-ESTIMATE HEADS 
C 
      E=E+ABS(H(NC,J)-G(NC)) 
      H(NC,J)=G(NC) 
      N=NC-1 
290   HA=G(N)-B(N)*H(N+1,J) 
      E=E+ABS(H(N,J)-HA) 
      H(N,J)=HA 
      N=N-1 
      IF(N)300,300,290 
300   CONTINUE 
      print *, e 
      IF(ITER.GT.100)GO TO 999 
      IF(E.GT.ERROR) GO TO 80 
C 
C          PRINT RESULTS 
C 
      WRITE(6,310)TIME,E,ITER, H(16,17), H(16,26) 
310   FORMAT(2F10.3,I5, 2F10.3) 
320   DELTA=DELTA*1.2 
C      DO 320 J=1,NR 
C320   WRITE(6,330)J,(H(I,J),I=1,NC) 
C330   FORMAT(I5,5X,10F10.4/(12X10F10.4)) 
      STOP 
C 
  999 WRITE(6,307) 
  307 FORMAT(//'*** ITERATION LIMIT EXCEEDED ***') 
      STOP 
      END 


